The Brouncker algorithm for the continued fractions of a root of a quadratic.

```
> with(linalg):
```

1 Basic calculations and functions.

To start, we create a matrix from $x^2 - 7\,y^2$

```
> A:=matrix(2,4,[1,0,1,0,0,1,0,-7]);
```

$$A := \begin{bmatrix} 1 & 0 & 1 & 0 \\ 0 & 1 & 0 & -7 \end{bmatrix}$$

The function "fun" produces a polynomial in x,y from a 2 x 2 matrix. We need its values to calculate steps.

```
> fun:=m-> unapply(m[1,3]*x^2+2*m[1,4]*x*y+m[2,4]*y^2,x,y);
```

$$fun := m \to \text{unapply}(x^2\, m_{1,3} + 2\,x\,y\,m_{1,4} + y^2\, m_{2,4},\, x,\, y)$$

```
> fun(A)(x,y);fun(A)(3,1);
```

$$x^2 - 7\,y^2$$
$$2$$

The function ch(m,t,n) changes matrix m as follows:

If n=1 then it add t times row 1 to row 2.

If n=2 it adds t times row 2 to row 1.

Then it makes a symmetric matrix operation on the last two columns.

```
> ch:=proc(m,t,n) local tmp:
> tmp:=copy(m); if n=1 then  tmp:=addrow(tmp,1,2,t);
> tmp:=addcol(tmp,3,4,t);
> elif n=2 then
> tmp:=addrow(tmp,2,1,t);tmp:=addcol(tmp,4,3,t);fi;op(tmp);end:
```

2 Case of $\sqrt{7}$

We start work on $x^2 - 7\,y^2$ **.**

We shall record the t-values as we proceed.

```
> A:=matrix(2,4,[1,0,1,0,0,1,0,-7]);
```

$$A := \begin{bmatrix} 1 & 0 & 1 & 0 \\ 0 & 1 & 0 & -7 \end{bmatrix}$$

We check the function at (1,1);

If it is negative, then we try values at (t,1) and stop just before the values change sign.

```
> fun(A)(1,1); fun(A)(2,1);fun(A)(3,1);
```

$$-6$$
$$-3$$
$$2$$

So set t=2 and n=1(because the value at (1,1) was less than 1.)

```
> tvals:=[2];
```

$$tvals := [2]$$

```
> A1:=ch(A,2,1);
```

$$A1 := \begin{bmatrix} 1 & 0 & 1 & 2 \\ 2 & 1 & 2 & -3 \end{bmatrix}$$

Repeat with new matrix. Here, the value at (1,1) is bigger than 1, so we try (1,t), until sign changes.

```
> fun(A1)(1,1);fun(A1)(1,2);
```

$$2$$
$$-3$$

So we set t=1 and n=2 (because the value at (1,1) was bigger than 1.)

```
> tvals:=[op(tvals),1];
```

$$tvals := [2, 1]$$

```
> A2:=ch(A1,1,2);
```

$$A2 := \begin{bmatrix} 3 & 1 & 2 & -1 \\ 2 & 1 & -1 & -3 \end{bmatrix}$$

Repeat! Again, we get t=1, n=1.

```
> fun(A2)(1,1);fun(A2)(2,1);
```

$$-3$$

$$1$$

```
> tvals:=[op(tvals),1];
```

$$tvals := [2, 1, 1]$$

```
> A3:=ch(A2,1,1);
```

$$A3 := \begin{bmatrix} 3 & 1 & 2 & 1 \\ 5 & 2 & 1 & -3 \end{bmatrix}$$

```
> fun(A3)(1,1);
```

$$1$$

This is a special case. When the function evalutes to 1, then set t=1 and reverse the order of row operations (i.e. change n from 1 to 2 or 2 to 1.

```
> tvals:=[op(tvals),1];
```

$$tvals := [2, 1, 1, 1]$$

```
> A4:=ch(A3,1,2);
```

$$A4 := \begin{bmatrix} 8 & 3 & 1 & -2 \\ 5 & 2 & -2 & -3 \end{bmatrix}$$

```
> fun(A4)(1,1);fun(A4)(2,1);fun(A4)(3,1);fun(A4)(4,1);fun(A4)(5,1);
```

$$-6$$

$$-7$$

$$-6$$

$$-3$$

$$2$$

Now we get t=4, n=1.

```
> tvals:=[op(tvals),4];
```

$$tvals := [2, 1, 1, 1, 4]$$

```
> A5:=ch(A4,4,1);
```

$$A5 := \begin{bmatrix} 8 & 3 & 1 & 2 \\ 37 & 14 & 2 & -3 \end{bmatrix}$$

```
> seq(cat(A,i)=op(cat(A,i)),i=1..5);
```

$$A1 = \begin{bmatrix} 1 & 0 & 1 & 2 \\ 2 & 1 & 2 & -3 \end{bmatrix},\ A2 = \begin{bmatrix} 3 & 1 & 2 & -1 \\ 2 & 1 & -1 & -3 \end{bmatrix},\ A3 = \begin{bmatrix} 3 & 1 & 2 & 1 \\ 5 & 2 & 1 & -3 \end{bmatrix},\ A4 = \begin{bmatrix} 8 & 3 & 1 & -2 \\ 5 & 2 & -2 & -3 \end{bmatrix},$$

$$A5 = \begin{bmatrix} 8 & 3 & 1 & 2 \\ 37 & 14 & 2 & -3 \end{bmatrix}$$

Note that A1 and A4 have arepeat of the second part. So, now the pattern repeats!

```
> convert(sqrt(7), confrac,10,'ans7');
```

$$[2, 1, 1, 1, 4, 1, 1, 1, 4, 1]$$

```
> ans7;
```

$$[2, 3, \frac{5}{2}, \frac{8}{3}, \frac{37}{14}, \frac{45}{17}, \frac{82}{31}, \frac{127}{48}, \frac{590}{223}, \frac{717}{271}]$$

```
> evalf(ans7);
```

[2., 3., 2.500000000, 2.666666667, 2.642857143, 2.647058824, 2.645161290, 2.645833333, 2.645739910, 2.645756458]

```
> seq(evalf[20](sqrt(7)-ans7[i]),i=1..nops(ans7));
```

0.6457513110645905905, −0.3542486889354094095, 0.1457513110645905905, −0.0209153556020760762, 0.0028941682074477334, −0.0013075124648211742, 0.0005900207420099453, −0.0000820222687427428, 0.0000114007506892452, $-0.51464999850553\,10^{-5}$

3 Case of $\sqrt{12}$

We set D=12. This should be non square.

Starting matrix

```
> A:=matrix(2,4,[1,0,1,0,0,1,0,-12]);tvals:=[];
```

$$A := \begin{bmatrix} 1 & 0 & 1 & 0 \\ 0 & 1 & 0 & -12 \end{bmatrix}$$

$$tvals := []$$

Check the value of fun at (1,1).

```
> fun(A)(x,y);fun(A)(1,1);
```

$$x^2 - 12\,y^2$$

$$-11$$

For a negative value (or < 1), we increase the first coordinate until sign changes. Add the tvalue just before the change to the list.

```
> seq([t,fun(A)(t,1)],t=1..5);
```

$$[1, -11], [2, -8], [3, -3], [4, 4], [5, 13]$$

Thus 3 is the turning point.

```
> tvals:=[op(tvals),3];
```

$$tvals := [3]$$

ch does the necessary row/col operations.

```
> A1:=ch(A,3,1);
```

$$A1 := \begin{bmatrix} 1 & 0 & 1 & 3 \\ 3 & 1 & 3 & -3 \end{bmatrix}$$

```
> fun(A1)(1,1);
```

$$4$$

Since the value is bigger than 1, we increase the second coordinate.

```
> seq([t,fun(A1)(1,t)],t=1..5);
```

$$[1, 4], [2, 1], [3, -8], [4, -23], [5, -44]$$

```
> tvals:=[op(tvals),2];A2:=ch(A1,2,2);
```

$$tvals := [3, 2]$$

$$A2 := \begin{bmatrix} 7 & 2 & 1 & -3 \\ 3 & 1 & -3 & -3 \end{bmatrix}$$

```
> fun(A2)(1,1);
```

$$-8$$

```
> seq([t,fun(A2)(t,1)],t=1..10);
```

$$[1, -8], [2, -11], [3, -12], [4, -11], [5, -8], [6, -3], [7, 4], [8, 13], [9, 24], [10, 37]$$

```
> tvals:=[op(tvals),6];A3:=ch(A2,6,1);
```

$$tvals := [3, 2, 6]$$

$$A3 := \begin{bmatrix} 7 & 2 & 1 & 3 \\ 45 & 13 & 3 & -3 \end{bmatrix}$$

```
> fun(A3)(1,1);
```

$$4$$

```
> seq([t,fun(A3)(1,t)],t=1..5);
```

$$[1, 4], [2, 1], [3, -8], [4, -23], [5, -44]$$

```
> tvals:=[op(tvals),2];
```

$$tvals := [3, 2, 6, 2]$$

```
> A4:=ch(A3,2,2);
```

$$A4 := \begin{bmatrix} 97 & 28 & 1 & -3 \\ 45 & 13 & -3 & -3 \end{bmatrix}$$

$$13$$

```
> seq(cat(A,i)=op(cat(A,i)),i=1..4);
```

$$A1 = \begin{bmatrix} 1 & 0 & 1 & 3 \\ 3 & 1 & 3 & -3 \end{bmatrix}, A2 = \begin{bmatrix} 7 & 2 & 1 & -3 \\ 3 & 1 & -3 & -3 \end{bmatrix}, A3 = \begin{bmatrix} 7 & 2 & 1 & 3 \\ 45 & 13 & 3 & -3 \end{bmatrix}, A4 = \begin{bmatrix} 97 & 28 & 1 & -3 \\ 45 & 13 & -3 & -3 \end{bmatrix}$$

```
> tvals;
```

$$[3, 2, 6, 2]$$

```
Note that A4 and A2 share the same RHS.  So the sequence [2,6] will
repeat!

\textbf{{\large Doublecheck using Maple directly!}}
> unassign('ans12');convert(sqrt(12),confrac,12,ans12);# calculates
12
> "a"s.
```

$$[3, 2, 6, 2, 6, 2, 6, 2, 6, 2, 6, 2]$$

```
> ans12;# successive continued fractions.
```

$$[3, \frac{7}{2}, \frac{45}{13}, \frac{97}{28}, \frac{627}{181}, \frac{1351}{390}, \frac{8733}{2521}, \frac{18817}{5432}, \frac{121635}{35113}, \frac{262087}{75658}, \frac{1694157}{489061}, \frac{3650401}{1053780}]$$

```
> evalf(%);#evaluations.
```

[3., 3.500000000, 3.461538462, 3.464285714, 3.464088398, 3.464102564, 3.464101547, 3.464101620, 3.464101615, 3.464101615, 3.464101615, 3.464101615]

```
> evalf[20](sqrt(12));
```

$$3.4641016151377545870$$

```
> seq(evalf[20](sqrt(12)-ans12[i]),i=1..12);
```

$0.4641016151377545870, -0.0358983848622454130, 0.0025631535992930485, -0.0001840991479596987, 0.0000132173476993384, -0.9489648095156\,10^{-6}, 0.681325979031\,10^{-7}, -0.48917004940\,10^{-8}, 0.3512082936\,10^{-9}, -0.252156210\,10^{-10}, 0.18104001\,10^{-11}, -0.1299809\,10^{-12}$